
Данная статья продолжает тему об использовании цифровых синтезаторов 

сигналов для создания высокочастотных перестраиваемых генераторов. В статье 

рассматривается генератор синусоидального сигнала на основе контроллера 

Arduino и синтезатора AD9850, предназначенный для настройки фильтров с 

резонаторами. Генератор работоспособен на частотах до 30 МГц. 

 

Цифровой синтезатор AD9850, выпускаемый корпорацией Analog Devices, 

способен формировать сигналы синусоидальной и прямоугольной формы с 

частотой от 0 до 62,5 МГц. Частота и фаза выходного сигнала AD9834 являются 

программируемыми: программирование осуществляется с внешнего 

микроконтроллера либо по параллельному, либо по последовательному 

интерфейсу. 

На рисунке 1 в качестве примера показана простая схема высокочастотного 

генератора синусоидального сигнала, построенного на основе микросхемы 

AD9850BRSZ, контроллера Arduino и инкрементального энкодера. Частота 

сигнала на выходе генератора определяется информацией, поступающей с 

энкодера на контроллер Arduino, программой, загруженной в этот контроллер, а 

также частотой тактового сигнала, подаваемого на микросхему DA1 синтезатора 

сигнала AD9850 с кварцевого генератора DD1. Напряжение питания +5В, 

необходимое для работы кварцевого генератора и синтезатора, обеспечивает 

стабилизатор, встроенный в контроллер Arduino. 

В данном примере используется тактовый сигнал с частотой 80 МГц. Код К 

выходной частоты F для синтезатора AD9850 в таком случае должен вычисляться 

по следующей формуле: 

 

K = F × 232 / 80000000. 

 

Результат вычислений должен быть округлен до 32-разрядного целого числа без 

знака, а затем загружен в регистр синтезатора, хранящий код частоты. 

Пример программы, позволяющей перестраивать частоту генератора с 

шагом 20 Гц в диапазоне от 9,996 до 10,004 МГц, приведен в листинге 1. 

Рисунок 1. Схема высокочастотного генератора на микросхеме AD9850 

 

Листинг 1. Пример программы, предназначенной для управления работой 

цифрового синтезатора сигналов AD9850 
 

#include <Arduino.h> 

#define W_CLK    8 // AD9850: W_CLK 

#define FQ_UD    9 // AD9850: FQ_UD 

#define RESET   10 // AD9850: RESET 

#define DATA    11 // AD9850: Serial data 

#define pin_CLK  2 // Энкодер: пин A 

#define pin_DT   4 // Энкодер: пин B 

#define PosLim 200 // Ограничитель изменения положения энкодера 

 

const float fgen = 80.0E6; // Частота сигнала ГТИ, Гц 

const float sf = 10.0E6;   // Начальное значение выходной частоты 

const float df = 50.0;     // Шаг приращения частоты, Гц 

 

volatile long Position = 0; // Текущее значение положения энкодера 

long oldPosition = 0;       // Предыдущее положение энкодера 

 

float ff;        // Код частоты в формате с плавающей точкой 

unsigned long f; // Двоичный код выходной частоты 

 

// Процедура для обработки прерывания от энкодера 

void EncoderRotate() { 

  if (digitalRead(pin_CLK) == digitalRead(pin_DT)) { 

    if(Position < PosLim) Position++; 



  } else { 

    if(Position > -PosLim) Position--; 

  } 

} 

 

// Процедура инициализации AD9850 

void AD9850init(){ 

  digitalWrite(RESET, HIGH); 

  digitalWrite(RESET, LOW); 

  digitalWrite(W_CLK, HIGH); 

  digitalWrite(W_CLK, LOW); 

  digitalWrite(FQ_UD, HIGH); 

  digitalWrite(FQ_UD, LOW); 

} 

 

// Процедура для передачи кода частоты синтезатору AD9850 

// по последовательному интерфейсу 

void AD9850setfreq(unsigned long Freq){ 

  int i; 

  unsigned long d; 

  d = Freq; 

  // Загрузка кода частоты 

  for(i=0; i<32; i++) 

  { 

    if(d & 1 != 0) digitalWrite(DATA, HIGH); 

    else digitalWrite(DATA, LOW); 

    digitalWrite(W_CLK, HIGH); 

    digitalWrite(W_CLK, LOW); 

    d = d >> 1; 

  } 

  // Загрузка управляющего байта, заполненного нулями 

  digitalWrite(DATA, LOW); 

  for(i=0; i<8; i++) 

  { 

    digitalWrite(W_CLK, HIGH); 

    digitalWrite(W_CLK, LOW); 

  } 

  // Завершение загрузки данных 

  digitalWrite(FQ_UD, HIGH); 

  digitalWrite(FQ_UD, LOW); 

} 

 

void setup() { 

  int i; 

  pinMode(W_CLK, OUTPUT); 

  pinMode(FQ_UD, OUTPUT); 

  pinMode(DATA, OUTPUT); 

  pinMode(RESET, OUTPUT); 

  pinMode(pin_CLK, INPUT); 

  pinMode(pin_DT, INPUT); 

 

  // Вычисляем начальный код частоты в формате с плавающей точкой 

  ff = sf*(float)0x10000l*(float)0x10000l/fgen; 



  // Получаем двоичный код выходной частоты 

  f = (unsigned long)ff; 

  // Инициализируем микросхему AD9850 

  AD9850init(); 

  // Загружаем код начального значения частоты 

  AD9850setfreq(f); 

   

  // Подключаем обработчик прерывания энкодера 

  attachInterrupt(digitalPinToInterrupt(pin_CLK), 

    EncoderRotate, RISING); 

} 

 

void loop() { 

  if (oldPosition != Position) 

  { 

    // Запомнаем текущую позицию 

    oldPosition = Position; 

    // Вычисляем код частоты в формате с плавающей точкой 

    ff=(sf+df*(float)Position)*(float)0x10000l*(float)0x10000l/fgen; 

    // Получаем двоичный код выходной частоты 

    f = (unsigned long)ff; 

    // Загружаем код частоты 

    AD9850setfreq(f); 

  } 

} 

 

При повороте ручки энкодера в ту или иную сторону генерируется 

прерывание и контроллер либо увеличивает, либо уменьшает значение 

переменной-счетчика положения энкодера Position. После любого изменения 

значения данной переменной контроллер заново вычисляет код частоты и 

загружает его в микросхему AD9850. 

Форма сигнала на выходе генератора при частоте колебаний 10 МГц 

показана на рисунке 2. Амплитуда выходного сигнала составляет 2 В. 

 

 

 
 

Рисунок 2. Осциллограмма сигнала на выходе генератора при частоте 10 МГц 



Осциллограмма, демонстрирующая результат применения быстрого 

преобразования Фурье к сигналу на выходе генератора, приведена на рисунке 3. 

 

  

 
 

Рисунок 3. Результат применения быстрого преобразования Фурье 

